// PROJECT  :BitsandBytes
// PURPOSE  :Preparation for Register-Level Coding
// DEVICE   :*
// AUTHOR   :C. D'Arcy
// DATE     :*
// MCU      :*
// COURSE   :ICSnU-E
// STATUS   :Working

char buff[60];  //large enough buffer to handle all output strings

void setup() {
  Serial.begin(9600);
  while (!Serial);

  //define an 8 bit value (reg is short for register)
  uint8_t reg = 0x07;  // 7(16) = 7(10) = B00000111

  // without leading zeroes...
  Serial.println("The value " + String(reg) + " in binary is " + String(reg, BIN));
  // with leading zeroes!
  sprintf(buff, "The value of %d in (full) binary %s.", reg, fullBinary(reg));
  Serial.println(buff);

  // turn on (set) a specific bit (aka make it 1): bitwise OR (|)
  uint8_t n = 5;
  reg |= 1 << n;
  sprintf(buff, "Setting bit %d yields %s.", n, fullBinary(reg));
  Serial.println(buff);

  // turn off (clear) a specific bit (aka make is 0): bitwise AND (&)
  n = 2;
  reg &= ~(1 << n);
  sprintf(buff, "Clearing bit %d yields %s.", n, fullBinary(reg));
  Serial.println(buff);

  //Setting multiple bits in a Register
  reg |= 1 << 7 | 1 << 4;
  sprintf(buff, "Setting bits simultaneously yields %s.", fullBinary(reg));
  Serial.println(buff);

  //Clearing multiple bits in a Register
  reg &= ~(1 << 5 | 1 << 4 | 1 << 1);
  sprintf(buff, "Clearing bits simultaneously yields %s.", fullBinary(reg));
  Serial.println(buff);
}

//Precondition: None
//PostCondition: The fully-formed 8-bit binary representation of the parameter is returned as a string
char* fullBinary(uint8_t value) {
  static char result[9] = "";
  for (uint8_t i = 0; i < 8; i++)
    result[7 - i] = value & 1 << i ? '1' : '0';
  return result;
}

